3. HIGH SCHOOL ALGEBRA

3.1 Polynomials and rational functions

3.1.1 Factoring a polynomial

MAPLE can do high school algebra. It can manipulate polynomials and rational functions of one or more variables quite easily.

```
> p := x^2+5*x+6;
```

$$p := x^2 + 5x + 6$$

```
> factor(p);
```

$$(x + 3)(x + 2)$$

```
> b := 1 - q^7 - q^8 - q^9 + q^15 + q^16 + q^17 - q^24;
```

$$b := 1 - q^7 - q^8 - q^9 + q^{15} + q^{16} + q^{17} - q^{24}$$

```
> factor(b);
```

$$-(q + 1)(q^2 + 1)(q^2 + q + 1)(q^6 + q^3 + 1)(q^4 + 1)$$
$$(q^6 + q^5 + q^4 + q^3 + q^2 + q + 1)(q - 1)^3$$

To factor a polynomial or rational function, we use `factor`. We let $p = x^2+5x+6$ and found the factorization using `factor(p)`. This could have easily been done by hand. Factoring $b = 1 - q^7 - q^8 - q^9 + q^{15} + q^{16} + q^{17} - q^{24}$ is not so easy, but child's play for MAPLE.

We can also use a context menu to factor a polynomial.

```
> p;
```

$$x^2 + 5x + 6$$

Use the right mouse button to click on the polynomial. A context menu should appear. Select `Factor`.

```
> R0 := factor(x^2+5*x+6);
```

$$R0 := (x + 3)(x + 2)$$

3.1.2 Expanding an expression

To expand a polynomial use `expand`. The command `combine` is also useful for expanding certain expressions.

```
> p := (x+2)*(x+3);
```

$$(x+2)\,(x+3)$$

```
> expand(%);
```

$$x^2+5\,x+6$$

```
> (1-q^8)*(1-q^7)*(1-q^6);
```

$$\left(1-q^8\right)\left(1-q^7\right)\left(1-q^6\right)$$

```
> expand(%);
```

$$1-q^6-q^7+q^{13}-q^8+q^{14}+q^{15}-q^{21}$$

```
> y := sqrt(x+2)*sqrt(x+3);
```

$$\sqrt{x+2}\sqrt{x+3}$$

```
> expand(y);
```

$$\sqrt{x+2}\sqrt{x+3}$$

```
> combine(y);
```

$$\sqrt{x+2}\sqrt{x+3}$$

```
> combine(y,radical);
```

$$\sqrt{x+2}\sqrt{x+3}$$

```
> combine(y,radical,symbolic);
```

$$\sqrt{x^2+5\,x+6}$$

Notice we were not able to expand the expression $(x+2)^{1/2}(x+3)^{1/2}$ with `expand` and had to use `combine`, using two additional arguments, `radical` and `symbolic`.

3.1.3 Collecting like terms

In the last section y had the value $\sqrt{x+2}\,\sqrt{x+3}$.

```
> y;
```

$$\sqrt{x+2}\sqrt{x+3}$$

To start over we use the `restart` function.

```
> restart:
> y;
```

$$y$$

Now y is just y. See Section 3.1.10 for another way to restore y to its variable status. One can also restart by pressing the restart button  in the tool bar.

The `collect` function is useful when looking at a polynomial in more than one variable.

```
> (x+y+1)*(x-y+1)*(x-y-1);
```

$$(x+y+1)(x-y+1)(x-y-1)$$

```
> p := expand(%);
```

$$p := x^3 - x^2y + x^2 - 2\,xy - x - y^2x + y^3 + y^2 - y - 1$$

```
> collect(p,x);
```

$$x^3 + (1-y)\,x^2 + \left(-1 - y^2 - 2\,y\right)x - y - 1 + y^3 + y^2$$

We let $p = (x+y+1)(x-y+1)(x-y-1) = x^3 - x^2y + x^2 - 2\,xy - x - y^2x + y^3 + y^2 - y - 1$. We used `collect(p,x)` to write p as a polynomial in x with coefficients that were polynomials in the remaining variable y. Similarly, try `collect(p,y)` to get p as a polynomial in y.

3.1.4 Simplifying an expression

The first thing you should try when presented with a complicated expression is `simplify`.

```
> 3*4^(1/2)+5;
```

$$3\sqrt{4} + 5$$

```
> simplify(%);
```

$$11$$

```
> x^2;
```

$$x^2$$

```
> %^(1/2);
```

$$\sqrt{x^2}$$

```
> simplify(%);
```

$$\mathtt{csgn}(x)\,x$$

Notice we were able to simplify $3\sqrt{4}+5$ to 11. Of course, the value of $(x^2)^{1/2}$ depends on the sign of x. Here `csgn` is a function that returns 1 if x is positive and -1 otherwise. It is also defined for complex numbers. See `?csgn` for more information. If we know that $x > 0$, we can use `assume` to do further simplification (x~ replaces x).

```
> y:=((x-2)^2)^(1/2);
```

$$y := \sqrt{(x-2)^2}$$

```
> assume(x>2);
> simplify(y);
```

$$x\sim - 2$$

To show the assumptions placed on a variable, we use the `about` function.

```
> about(x);
Originally x, renamed x~:
  is assumed to be:  RealRange(Open(2),infinity)
```

The output `RealRange(Open(2),infinity)` means the interval $(2, \infty)$. This translates into the assumption that $x > 2$.

To remove the assumption on x, we could use the `restart` function, but then we would lose the value of y. Instead we do the following.

```
> x :='x';
```

$$x := x$$

This restores x to its original status. See Section 3.1.9.

MAPLE 7 has a nifty new command called `assuming`. This allows us to do simplifications with temporary assumptions.

```
> y;
```

$$\sqrt{(x-2)^2}$$

```
> simplify(y) assuming x>2;
```

$$x - 2$$

The last command simplified y under the assumption that $x > 2$. Notice that the output is in terms of x and not $x\sim$.

3.1.5 Simplifying radicals

To simplify expressions using radicals, we can use `simplify` and `radsimp`.

```
> y := x^3 + 3*x^2 + 3*x + 1;
```

$$y := x^3 + 3x^2 + 3x + 1$$

```
> simplify(y^(1/3));
```

$$((1+x)^3)^{1/3}$$

```
> radsimp(y^(1/3));
```

$$1 + x$$

```
> assume(x>-1);
```

```
> simplify(y^(1/3));
```

$$1 + x\sim$$

```
> assume(x<-1);
```

```
> simplify(y^(1/3));
```

$$-\frac{1}{2}(x\sim+1)(1+I\,3^{1/2})$$

```
> x := 'x':
```

Notice that `simplify` recognized y as a cube but failed to simplify $y^{1/3}$. The command `radsimp`, on the other hand, was able to simplify $y^{1/3}$ to $1+x$. If assumptions are given for x, then `simplify` is able to simplify the radical further. However, it should be noted that the value of the cube root depends on these assumptions, so care needs to be taken.

A cute MAPLE command is `rationalize`.

```
> 1/(1+sqrt(2));
```

$$\frac{1}{\sqrt{2}+1}$$

```
> rationalize(%);
```

$$\sqrt{2}-1$$

```
> (1-2^(2/3))/(1+2^(1/3));
```

$$\frac{1-2^{2/3}}{1+2^{1/3}}$$

```
> rationalize(%);
```

$$-2^{1/3}+1$$

```
> y:= z/(1 + sqrt(x));
```

$$y := \frac{z}{1+\sqrt{x}}$$

```
> rationalize(y);
```

$$\frac{z\,(-1+\sqrt{x})}{-1+x}$$

Notice that `rationalize` does a great job rationalizing a denominator not only for expressions involving square roots but for more complicated radicals as well. It can also handle symbolic expressions.

3.1.6 Working in the real domain

Sometimes MAPLE will return an expected complex number. We saw an instance of this in the last section. We reexamine the example.

```
> restart:
> y := (1+x)^3:
> simplify(y^(1/3)) assuming x<-1;
```

$$-\frac{1}{2}(x\sim+1)(1+I\,3^{1/2})$$

Here I is MAPLE's notation for the complex number $i = \sqrt{-1}$. The unsuspecting precalculus or calculus student may not be expecting this complex cube root of y and would prefer to work in the real domain. Fortunately, there is a new package in MAPLE 7 for working in the real domain. Funnily enough the package is called *RealDomain*. To load this package we must use the `with` function.

```
> with(RealDomain):
Warning,
these protected names have been redefined and unprotected:
Im, Re, ^, arccos, arccosh, arccot, arccoth, arccsc, arccsch,
arcsec, arcsech, arcsin, arcsinh, arctan, arctanh, cos, cosh, cot,
coth, csc, csch, eval, exp, expand, limit, ln, log, sec, sech,
signum, simplify, sin, sinh, solve, sqrt, surd, tan, tanh
```

We redo the calculation of $y^{1/3}$:

```
> y := (1+x)^3:
> simplify(y^(1/3)) assuming x<-1;
```

$$1 + x$$

```
> simplify(y^(1/3));
```

$$1 + x$$

This time $y^{1/3}$ simplified to $1+x$. This is the only real cube root of y, assuming x is real.

Let's redo some calculations from Section 3.1.4, but this time in the real domain.

```
> with(RealDomain):
> x^2;
```

$$x^2$$

```
> %^(1/2);
```

$$\sqrt{x^2}$$

```
> simplify(%);
```

$$|x|$$

```
> y:=x^3+3*x^2+3*x+1;
```

$$x^3 + 3\,x^2 + 3\,x + 1$$

```
> simplify(y^(1/3));
```

$$signum(x^3 + 3x^2 + 3x + 1)^{2/3}\,((x+1)^3)^{1/3}$$

This time in the real domain we found $\sqrt{x^2} = |x|$, which is more palatable than $\text{csgn}(x)x$. Here $y^{1/3}$ should have simplified to $x + 1$, so I guess MAPLE still is not perfect.

```
> restart:
> sqrt(-1);
```

$$I$$

```
> with(RealDomain):
> sqrt(-1);
```

$$\textit{undefined}$$

After restarting, MAPLE recognizes $\sqrt{-1}$ as the complex number i. When *RealDomain* is loaded, MAPLE considers $\sqrt{-1}$ as being undefined.

3.1.7 Simplifying rational functions

To simplify a rational function (i.e., a function that can be written as a quotient of two polynomials) we use the command `normal`. This has the effect of canceling any common factors between numerator and denominator. First we restore x and y's variable status.

```
> y:='y':  z:='z':
> a:= (x-y-z)*(x+y+z);
```

$$a := (x - y - z)(x + y + z);$$

```
> b :=(x^2-2*x*y-2*x*z+y^2+2*y*z+z^2)*(x^2-x*y+x*z-y*z);
```

$$b := (x^2 - 2xy - 2xz + y^2 + 2yz + z^2)(x^2 - xy + xz - yz)$$

```
> c:=a/b;
```

$$c := \frac{(x - y - z)(x + y + z)}{(x^2 - 2xy - 2xz + y^2 + 2yz + z^2)(x^2 - xy + xz - yz)}$$

```
> normal(c);
```

$$-\frac{(x + y + z)}{(x^2 - yx + xz - yz)(-x + y + z)}$$

```
> simplify(c);
```

$$-\frac{(x + y + z)}{(x^2 - yx + xz - yz)(-x + y + z)}$$

```
> factor(c);
```

$$\frac{(x + y + z)}{(x - y)(x + z)(x - y - z)}$$

Observe that `normal` and `simplify` had the same effect on the rational function c. We use `normal` for rational functions if we can do without the more expensive `simplify`. Also, we could have used `factor` to simplify c and get it into a nice form. It should be noted that `normal` is able to do this simplification without factoring, which is more expensive in terms of memory.

Some useful functions for manipulating rational functions are: `numer`, `denom`, `rem`, and `quo`. We let c be as above.

```
> numer(c);
```

$$-(-x+y+z)(x+y+z)$$

```
> denom(c);
```

$$(x^2-2xy-2xz+y^2+2yz+z^2)(x^2-xy+xz-yz)$$

```
> factor(%);
```

$$(-x+y+z)^2(x-y)(x+z)$$

The functions `numer` and `denom` select the numerator and denominator, respectively, of a rational function. After factoring the denominator of c, we see that there was simplification because of the common factor $(-x+y+z)$.

Many operations on rational functions can also be performed through a context menu.

```
> c;
```

$$\frac{(x-y-z)(x+y+z)}{(x^2-2xy-2xz+y^2+2yz+z^2)(x^2-xy+xz-yz)}$$

Click the right mouse button on our rational function above. A context menu should appear. Now try clicking on `Factor`, `Simplify`, `Expand`, `Normal`, `Numerator`, and `Denominator`.

The functions `quo` and `rem` give the quotient and remainder upon polynomial division.

```
> a:= 2*x^3+x^2+12;
```

$$a := 2x^3+x^2+12$$

```
> b := x^2 - 4;
```

$$b := x^2-4$$

```
> q := quo(a,b,x);
```

$$q := 2x+1$$

```
> r := rem(a,b,x);
```

$$r := 16+8x$$

```
> expand( a - (b*q + r) );
```

$$0$$

The command `quo(a,b,x)` gives the quotient q when a is divided by b as polynomials in x. The command `rem(a,b,x)` gives the remainder r so that

$$a = bq + r,$$

and the degree of r (as a polynomial in x) is less than the degree of b.

3.1.8 Degree and coefficients of a polynomial

In Section 3.1.3 the `collect` command was introduced to view polynomials. Two other useful functions are `coeff` and `degree`. Let p be as before.

```
> p:= y*(x+y+1)*(x-y+1)*(x-y-1):
> q := expand(%);
```

$$yx^3 - x^2y^2 + x^2y - 2\,y^2x - xy - y^3x + y^4 + y^3 - y^2 - y$$

```
> coeff(q,x,2);
```

$$-y^2 + y$$

```
> coeff(p,x,2);
```

$$y\,(y+1) + y\,(-y+1) + y\,(-y-1)$$

```
> expand(%);
```

$$-y^2 + y$$

```
> degree(q,x);
```

$$3$$

The command `coeff(q,x,2)` found the coefficient of x^2 in the polynomial q. The command `degree(q,x)` gave the degree of q as a polynomial in x. Observe also that when `coeff` was applied to the unexpanded form p, MAPLE still returned the correct value for the coefficient but in an unexpanded form.

Warning: In MAPLE V Release 4 (and earlier versions), `coeff` will either return an "incorrect" result or an error message, if it is applied to an unexpanded polynomial like p. So be careful when using `coeff` in these earlier versions of MAPLE.

Another useful and related function is `ldegree`.

```
> q := q - 2*x/y;
```

$$q := yx^3 - x^2y^2 + x^2y - 2\,y^2x - xy - y^3x + y^4 + y^3 - y^2 - y - 2\,\frac{x}{y}$$

```
> ldegree(q,x);
```

$$0$$

```
> ldegree(q,y);
```

$$-1$$

```
> c1:=1;
```

$$1$$

```
> c2:=0;
```

$$0$$

```
> degree(c1,x);
```

$$0$$

```
> degree(c2,x);
```

$$-\infty$$

The assignment `q := q - 2*x/y` subtracted $2x/y$ from q and assigned the result to q. `ldegree(q,x)` returns the degree of the lowest power of x in the polynomial q, which in our session was 0. Because of the term $2x/y$, `ldegree(q,y)` returned -1 as the lowest degree in the variable y. Also, observe that MAPLE returns 0 for the degree of a nonzero constant but returns $-\infty$ for the degree of the zero polynomial.

Warning: In MAPLE V Release 4 (and earlier versions), `degree` will return 0 for the zero polynomial.

3.1.9 Substituting into an expression

We can substitute into an expression using the command `subs`.

```
> p := (x+y+z)*(x-y+z)*(x-y-z);
```

$$p := (x + y + z)(x - y + z)(x - y - z)$$

```
> subs(x=1,p);
```

$$(1 + y + z)(1 - y + z)(1 - y - z)$$

To substitute $x = 1$ into p, we used the command `subs(x=1,p)`. Try substituting $x = 1$ and $y = 2$ into p using the command `subs(x=1,y=2,p)`.

3.1.10 Restoring variable status

In the last section we saw how `subs` is used to do substitution. There is another way to do this. We let p be as Section 3.1.8.

```
> p;
```

$$(x + y + z)(x - y + z)(x - y - z)$$

```
> x:=1:  y:=2:
> p;
```

$$(3 + z)(-1 + z)(-1 - z)$$

We are able to do the substitution by assigning $x := 1$ and $y := 2$. However, now p has changed. There is a way to restore x and y's variable status.

```
> x := 'x':   y := 'y':
> p;
```

$$(x + y + z)(x - y + z)(x - y - z)$$

The assignments `x := 'x'` and `y := 'y'` restored x and y to their variable status. It is neat that p was also restored to its original status.

3.2 Equations

3.2.1 Left- and right-hand sides

To assign a value to a variable, we use `:=`. The symbol `=` has a different meaning and is reserved for equations.

```
> eqn := x^2 - x = 1;
```

$$eqn := x^2 - x = 1$$

```
> R := solve(eqn,x);
```

$$R := \frac{1}{2}\sqrt{5} + \frac{1}{2}, \quad \frac{1}{2} - \frac{1}{2}\,5^{1/2}$$

```
> simplify(R[1]*R[2]);
```

$$-\frac{1}{4}\,(\sqrt{5}+1)\,(\sqrt{5}-1)$$

```
> expand(%);
```

$$-1$$

We assigned to equation $x^2 - x = 1$ the name *eqn*. We solved the equation for x by typing `solve(eqn,x)`. We named the list of solutions R. The two solutions were $R[1]$ and $R[2]$. In this way we can manipulate the solutions. Observe that we computed the product of the roots to be -1 as expected.

The left and right sides of an equation can be manipulated using `lhs` and `rhs`.

```
> eqn;
```

$$x^2 - x = 1$$

```
> lhs(eqn);
```

$$x^2 - x$$

```
> subs(x=R[1],lhs(eqn));
```

$$\left(\frac{1}{2} + \frac{1}{2}\,\sqrt{5}\right)^2 - \frac{1}{2}\,\sqrt{5} - \frac{1}{2}$$

```
> expand(%);
```

$$1$$

The command `lhs(eqn)` gave us the left side of the equation. Then we were able to substitute $x = R[1]$ (the first root) into the left side of the equation, which simplified to 1 (as expected) using `expand`.

3.2.2 Finding exact solutions

MAPLE has the capability of solving systems of equations.

```
> restart:
> eqn1 := x^3+a*x=14;
```

$$eqn1 := x^3 + ax = 14$$

```
> eqn2 := a^2-x=7;
```

$$eqn2 := a^2 - x = 7$$

```
> solve({eqn1,eqn2},{x,a});
```

$$\{a = 3, x = 2\},$$
$$\{a = RootOf(\_Z^5 + 3\,\_Z^4 - 12\,\_Z^3 - 35\,\_Z^2 + 42\,\_Z + 119, label = \_L1),$$
$$x = \left(RootOf(\_Z^5 + 3\,\_Z^4 - 12\,\_Z^3 - 35\,\_Z^2 + 42\,\_Z + 119, label = \_L1)\right)^2 - 7\}$$

The syntax for solving systems of equations is `solve(S,X)` where S is a set of equations and X is the required set of variables. Observe that MAPLE was able to find the solution $x = 2$, $a = 3$. It also found that $a = z$, $x = z^2 - 7$ are solutions where z is any root of the following polynomial equation:

$$Z^5 + 3Z^4 - 12Z^3 - 35Z^2 + 42Z + 119 = 0.$$

The argument `label = _L1` gives the root a label. This is a way of distinguishing roots when using the `RootOf` function. As in the previous section, we can manipulate solutions. We select the first set of solutions and substitute them into the first equation.

```
> %[1];
```

$$\{a = 3, x = 2\}$$

```
> subs(%,eqn1);
```

$$14 = 14$$

3.2.3 Finding approximate solutions

In the last section we came upon the following quintic:

$$Z^5 + 3Z^4 - 12Z^3 - 35Z^2 + 42Z + 119 = 0.$$

Although naturally enough MAPLE is unable to find an exact explicit solution, it is able to find approximate solutions using `fsolve`.

```
> polyeqn := Z^5+3*Z^4-12*Z^3-35*Z^2+42*Z+119=0:
> a1 := fsolve(polyeqn,Z);
```

$$a1 := -3.136896207$$

```
> x1:= a1^2 -7;
```

$$x1 := 2.840117813$$

```
> subs({x=x1,a=a1},{eqn1,eqn2});
```

$$\{14.00000003 = 14,\ 7.0000000000 = 7\}$$

We used the command `fsolve(polyeqn,Z)` to find the approximate solution $Z \approx -3.136896207$. This implied that $a = -3.136896207$ and $x = a^2 - 7 = 2.840117813$ are approximate solutions to our system of equations in the previous section. We were able to check this using `subs`.

3.2.4 Assigning solutions

Once an equation or system of equations has been solved, we can use `assign` to assign a particular solution to the variable(s). We use the example given in Section 3.2.2.

```
> solve({x^3+a*x=14,a^2-x=7},{a,x}):
> %[1];
```

$$\{a = 3, x = 2\}$$

```
> assign(%);
> a; x;
```

$$3$$

$$2$$

To restore a and x to variable status we could use the method of Section 3.1.9 or use the `unassign` function.

```
> unassign('a','x');
> a,x;
```

$$a, \quad x$$

3.3 Fun with integers

3.3.1 Complete integer factorization

The command `ifactor` gives the prime factorization of an integer.

```
> 2^(2^5)+1;
```

$$4294967297$$

```
> ifactor(%);
```

$$(641)\,(6700417)$$

```
> ifactor(500326623506762117579);
```

$$(3)^2\,(13)\,(31)^3\,(67)\,(139)\,(320057)\,(481577)$$

3.3.2 Quotient and remainder

The integer analogs of quo and rem, the functions for finding the quotient and the remainder in polynomial division, are the functions iquo and irem. They work in the same way.

```
> a := 23;   b := 5;
```

$$a := 23$$
$$b := 5$$

```
> q := iquo(a,b);  r := irem(a,b);
```

$$q := 4$$
$$r := 3$$

```
> b*q+r;
```

$$23$$

We observe that if $q =$ iquo(a,b) and $r =$ irem(a,b), then

$$a = bq + r,$$

where $0 \leq r < b$ if a and b are positive.

Two related functions are floor and frac. The function floor(x) gives the greatest integer less than or equal to x and frac(x) gives the fractional part of x. Try

```
> x := 22/7;
> floor(x);
> frac(x);
> floor(-x);
> frac(-x);
```

3.3.3 Gcd and lcm

The greatest common divisor and the lowest common multiple of a set of numbers can be found using gcd and lcm.

```
> gcd(28743,552805);
```

$$11$$

```
> ifactor(28743);   ifactor(552805);
```

$$(3)(11)(13)(67)$$
$$(5)(11)(19)(23)^2$$

```
> lcm(21,35,99);
```

$$3465$$

We find that the gcd of 28743 and 552805 is 11. This can also be seen from the prime factorizations. The lcm of 21, 35, and 99 is 3465.

3.3.4 Primes

The ith prime can be computed with `ithprime`. The function `isprime` tests whether a given integer is prime or composite.

```
> ithprime(100);
```

$$541$$

```
> isprime(2^101-1);
```

$$\mathit{false}$$

```
> 7*3^10 + 10;
```

$$413353$$

```
> isprime(%);
```

$$\mathit{true}$$

We found that the 100th prime is 541, that $2^{101} - 1$ is composite, and that $7 \cdot 3^{10} + 10 = 413353$ is prime. Try making a table of the first 200 primes:

```
> matrix(20,10,[seq(ithprime(k),k=1..200)]);
```

For a positive integer n, `nextprime(`n`)` gives the smallest prime larger than n, and `prevprime(`n`)` gives the largest prime smaller than n.

```
> nextprime(1000);
```

$$1009$$

```
> prevprime(1000);
```

$$997$$

The next prime past 1000 is 1009 and the previous prime is 997.

3.3.5 Integer solutions

In Sections 3.2.1 and 3.2.2 we saw how to solve equations in MAPLE using `solve`. The integer analog of `solve` is `isolve`. We use this function if we are only interested in integer solutions. We use the example from Section 3.2.2. Remember to restore variable status to x and a first.

```
> x:='x':  a:='a':
> eqn1:= x^3+a*x=14:  eqn2 := a^2-x=7:
> isolve({eqn1,eqn2},{x,a});
```

$$\{a = 3, x = 2\}$$

This time we found the unique integer solution $a = 3$, $x = 2$ to the given system of equations.

3.3.6 Reduction mod p

MAPLE can do computations with integers modulo m.

```
> modp(117,13);
```

$$0$$

```
> modp(129,13);
```

$$12$$

```
> ifactor(129-12);
```

$$(3)^2\,(13)$$

```
> 117 mod 13;
```

$$0$$

```
> 129 mod 13;
```

$$12$$

```
> 1/17 mod 257;
```

$$121$$

```
> modp(121*17,257);
```

$$1$$

The functions for reduction modulo m are `modp` and `mod`. Given an integer a and a positive integer m, `modp(a,m)` reduces a modulo m. The syntax using `mod` is `a mod m`. In our MAPLE session, `modp(129,13)` returned 12, which means

$$129 \equiv 12 \pmod{13},$$

and this is indeed the case in as much as 13 divides the difference $129 - 12$. The call `129 mod 13` also reduced 129 modulo 13. When a and m are relatively prime, i.e., 1 is their greatest common divisor, `modp(1/a,m)` or `1/a mod m` returns the multiplicative inverse of a modulo m. We see that 121 is the inverse of 17 modulo 257, and indeed

$$(121)(17) \equiv 1 \pmod{257}.$$

3.4 Unit conversion

MAPLE 7 has new facilities for converting from one system of units to another. There are both command line and menu-driven facilities. In the tool bar click on Edit and then on Unit Converter. A **Unit Converter** window should open.



Figure 3.1 Menu-driven unit converter.

The window is already set up to do a simple example. Notice that 1.0 (Fig. 3.1) is in the Value box, Dimension is set to length, and we are ready to do a conversion from feet to meters. Click on `Insert`.

```
>  convert( 1.0, 'units', 'ft', 'm' );
```

$$.3048000000$$

This means that

$$1.0\,\text{ft} = 0.3048\,\text{m}.$$

Let's try another conversion. Click on  in the Dimension box and select `temperature`. Notice that the units have changed in the From and To boxes. Let's convert 100 degrees Fahrenheit to degrees Celsius. In the Value box type `100.0`, select `degrees Celsius (degC)` in the To box, and press `Insert`.

```
>  convert( 100.0, 'temperature', 'degF', 'degC' );
```

$$37.7777778$$

This means that

$$100.0^{o}F \approx 37.778^{o}C.$$

We have seen two types of dimensions: length and temperature. There are many other dimensions available, including acceleration, angle, area, electric capacitance, force, magnetic flux, mass, power, pressure, speed, time, torque, volume, and work. A list of all dimensions can be obtained by loading the *Units* package and calling the `GetDimensions` function. Try

```
>  with(Units):
>  GetDimensions();
```

MAPLE 7 knows many systems of units, including SI, FPS, MKS, and CGS. See `?Units[System]` for more information.

The `convert` function can also be used to make conversion tables. We make a conversion table for meters, yards, kilometers, and miles:

```
> convert([m,yd,km,mile],conversion_table,output=grid,
   filter=evalf[6]);
```

		To :	m	yd	km	mi
Unit Name	Symbol					
meters	m		1.	1.09361	0.001	0.000621371
yards	yd		0.9144	1.	0.0009144	0.000568182
kilometers	km		1000.	1093.61	1.	.621371
miles	mi		1609.34	1760.	1.60934	1.

Here `evalf[6]` means to use `evalf` with 6 digits. From the table we see that to convert from miles to kilometers just multiply by 1.60934. For other examples see `?conversion,conversion_table`.

We can do MAPLE calculations using units. As an example we sum 12.1 feet and 4 meters.

```
> 12.1*Unit(ft)+4*Unit(m);
```

$$12.1\,[ft] + 4\,[m]$$

We can simplify this by loading the *Standard* function in the *Units* package:

```
> with(Units[Standard]):
> 12.1*Unit(ft)+4*Unit(m);
```

$$7.688080000\,[m]$$

This means that the sum of 12.1 feet and 4 meters is 7.68808 meters. Other MAPLE functions recognize these units.

```
> max(12.1*Unit(ft),4*Unit(m));
```

$$4\,[m]$$

This means that 4 meters is bigger than 12.1 feet. A different style of representing units can be used by loading the `Natural` function in the *Units* package. Try

```
> with(Units[Natural]):
> max(12.1*ft,4*m);
```

3.5 Trigonometry

3.5.1 Degrees and radians

To convert between degrees and radians we use `convert`.

```
> convert(72*degrees,radians);
```

$$2/5\,\pi$$

```
> convert(2/5*Pi,degrees);
```

$$72\,degrees$$

To convert d degrees to radians we use `convert(d*degrees,radians)`. To convert r (in radians) to degrees we use `convert(r,degrees)`. We see that 72 degrees is $2\pi/5$ radians. Remember, we use `Pi` for π in MAPLE. Alternatively, we could convert degrees to radians by multiplying by $\pi/180$.

```
> 72*Pi/180;
```

$$2/5\,\pi$$

In MAPLE 7 we can use the `convert` function with the `units` option as we did in Section 3.4. Try

```
> convert(72,units,degrees,radians);
> convert(2*Pi/5,units,radians,degrees);
```

3.5.2 Trigonometric functions

In MAPLE, the trigonometric functions are `sin`, `cos`, `tan`, `sec`, `csc`, and `cot`. The arguments for all the trigonometric functions are in radians.

```
> sin(0);
```

$$0$$

```
> cos(0);
```

$$1$$

```
> tan(0);
```

$$0$$

```
> sin(Pi/2);
```

$$1$$

```
> cos(Pi/2);
```

$$0$$

```
> tan(Pi/2);
Error, (in tan) singularity encountered
> cot(Pi/2);
```

$$0$$

Remember, $\tan(\pi/2)$ is not defined.

The inverse trigonometric functions are `arcsin` `arccos`, `arctan`, `arcsec`, `arccsc`, and `arccot`.

```
> arcsin(1/2);
```

$$1/6\,\pi$$

```
> arcsec(-2);
```

$$2/3\,\pi$$

```
> arctan(1);
```

$$1/4\,\pi$$

```
> arcsin(sin(Pi/12));
```

$$1/12\,\pi$$

```
> arcsin(sin(Pi/12+Pi));
```

$$-1/12\,\pi$$

We found that

$$\begin{aligned} \sin^{-1}(1/2) &= \pi/6, & \sec^{-1}(-2) &= 2\pi/3 \\ \tan^{-1}(1/2) &= \pi/4, & \sin^{-1}(\sin(\pi/12)) &= \pi/12 \\ \sin^{-1}(\sin(13\pi/12)) &= -\pi/12 \end{aligned}$$

3.5.3 Simplifying trigonometric functions

Ever have trouble remembering the addition formulas for the trigonometric functions? Try the following:

```
> expand(sin(a+b));
```

$$\sin(a)\cos(b) + \cos(a)\sin(b)$$

```
> expand(cos(a+b));
```

$$\cos(a)\cos(b) - \sin(a)\sin(b)$$

```
> expand(tan(a+b));
```

$$\frac{\tan(a) + \tan(b)}{1 - \tan(a)\tan(b)}$$

Now it all comes back to us:

$$\begin{aligned} \sin(a+b) &= \sin(a)\cos(b) + \cos(a)\sin(b) \\ \cos(a+b) &= \cos(a)\cos(b) - \sin(a)\sin(b) \\ \tan(a+b) &= \frac{\tan(a) + \tan(b)}{1 - \tan(a)\tan(b)} \end{aligned}$$

To simplify a trigonometric expression, use `simplify`.

```
> y:=(1+sin(x)+cos(x))/(1+sin(x)-cos(x));
```

$$\frac{1+\sin(x)+\cos(x)}{1+\sin(x)-\cos(x)}$$

```
> simplify(y);
```

$$-\frac{\sin(x)}{\cos(x)-1}$$

We found that

$$\frac{1+\sin(x)+\cos(x)}{1+\sin(x)-\cos(x)} = \frac{\sin(x)}{1-\cos(x)}.$$

Can you show this result by hand?

Now try the following:

```
> expand(sin(5*x));
```

$$16\,\sin(x)\,(\cos(x))^4 - 12\,\sin(x)\,(\cos(x))^2 + \sin(x)$$

```
> factor(%);
```

$$\sin(x)\left(4\,(\cos(x))^2 + 2\,\cos(x) - 1\right)\left(4\,(\cos(x))^2 - 2\,\cos(x) - 1\right)$$

This means that

$$\sin 5x = \sin x\,(4\cos^2 x + 2\cos x - 1)(4\cos^2 x - 2\cos x - 1).$$

By letting $x = \frac{2\pi}{5} = 72^{\circ}$, derive a nice value for $\cos 72^{\circ}$.